Physics 151
Chapter 13 Notes
Gravitational Orbits

13 Gravitation

13.1 Newton's Law of Gravitation

An object in a circular orbit is really in a state of continually falling, or "Freefall."



In some sense, gravity is nature's weakest force, but since it is always attractive, it's often the one we most notice.

The text just asserts the law and then proceeds, but we can also infer it from data on planetary orbits on the inside back cover of the textbook. *Mathematica* also has a lot of the relevant data built in.

Consider a planet in a circular orbit of radius R around the Sun. The planet takes a time T (known as the period) to complete one orbit. The orbital velocity $v = 2\pi R/T$. We can explore how these quantities are related.



Planetary Orbits

Mathematica has some basic planetary data built in. We can use it to re-discover Kepler's third law, and, from that, infer the form of Newton's law of Universal Gravitation. Since the orbits are not quite circular, the appropriate quantity to use is the semi-major axis (i.e. half of the major axis of the orbital ellipse.)

In[26]:=
```
Clear["Global`*"];
```

In[27]:=
```
planets =
  {"Mercury", "Venus", "Earth", "Mars", "Jupiter", "Saturn", "Uranus", "Neptune"};
```

In[28]:=
```
data = Table[{PlanetData[p, "MajorAxis"] / 2.0,
    PlanetData[p, "OrbitPeriod"]}, {p, planets}];
```

Pluto is technically classified as a "Dwarf Planet", so we have to add it in separately.

In[29]:=
```
AppendTo[planets, "Pluto"];
AppendTo[data, {MinorPlanetData["Pluto", "MajorAxis"] / 2.0,
  MinorPlanetData["Pluto", "OrbitPeriod"]}];
```

In[30]:=
```
TableForm[data,
 TableHeadings → {planets, {"Orbital Radius", "Orbital Period"}}]
```

Out[30]//TableForm=

	Orbital Radius	Orbital Period
Mercury	0.387099 au	87.96926 days
Venus	0.723332 au	224.70080 days
Earth	1. au	365.25636 days
Mars	1.52366 au	1.8808476 a
Jupiter	5.20336 au	11.862615 a
Saturn	9.53707 au	29.447498 a
Uranus	19.1913 au	84.016846 a
Neptune	30.069 au	164.79132 a
Pluto	39.5886 au	249. a

For plotting and computation, we should switch all data to uniform units of meters and seconds.

In[31]:=
```
data = UnitConvert[data, "SIBase"];
TableForm[data,
 TableHeadings → {planets, {"Orbital Radius", "Orbital Period"}}]
```

Out[32]//TableForm=

	Orbital Radius	Orbital Period
Mercury	5.79092×10^{10} m	7.600544×10^{6} s
Venus	1.08209×10^{11} m	1.9414149×10^{7} s
Earth	1.49598×10^{11} m	3.1558149×10^{7} s
Mars	2.27937×10^{11} m	5.9355036×10^{7} s
Jupiter	7.78412×10^{11} m	3.7435566×10^{8} s
Saturn	1.42673×10^{12} m	9.2929236×10^{8} s
Uranus	2.87097×10^{12} m	2.6513700×10^{9} s
Neptune	4.49825×10^{12} m	5.2004186×10^{9} s
Pluto	5.92237×10^{12} m	7.86×10^{9} s

Animation (inner planets only)

For animation, it is convenient to scale all lengths by Earth's orbital radius, and all times by Earth's orbital period (1 year):

In[33]:=
```
comp =
  Table[{data[[i, 1]] / data[[3, 1]], data[[i, 2]] / data[[3, 2]]}, {i, 1, Length[data]}];
```

In[34]:=
```
x[r_, T_, t_] := r Cos[2 π t / T]
y[r_, T_, t_] := r Sin[2 π t / T]
pos[r_, T_, t_] := {x[r, T, t], y[r, T, t]}
```

In[37]:=
```
planets = {1, 2, 3, 4};  (* Just show the 4 inner planets *)
Animate[Show[{
   Graphics[{Yellow, Disk[{0, 0}, 0.1]}] (* Sun *),
   ParametricPlot[
    Table[pos[comp〚i, 1〛, 1, ϕ], {i, planets}], {ϕ, 0, 1}, PlotRange → All],
   ListPlot[Table[{pos[comp〚i, 1〛, comp〚i, 2〛, t]}, {i, planets}],
    PlotStyle → {{Brown, PointSize[0.04]}, {Gray, PointSize[0.05]},
      {Blue, PointSize[0.06]}, {Red, PointSize[0.04]}
      (*, {Brown, PointSize[0.10]}, {Yellow, PointSize[0.08]} *)}]
  }
 ], {t, 0, 5, 0.01, DefaultDuration → 60, AnimationRunning → False}]
```

Out[38]=



In[39]:=
```
opts = {AspectRatio → 1 / 2, LabelStyle → Larger, Frame → True,
   FrameLabel → {"Mean Radius (m)", "Orbital Period (s)"},
   ImageSize → Scaled[0.75]}; (* Common plot options *)
```

In[40]:=
```
ListPlot[data, opts]
```
Out[40]=



This plot is not terribly useful for visualization, since all of the inner planets are squished down in the lower left corner, and it doesn't follow any obvious curve.

Power-Laws and Log-Log Plots

The T vs. R plot is not terribly useful for visualization, since all of the inner planets are squished down in the lower left corner. Using a Log-Log plot spreads things out nicely.

In[41]:=
```
ListLogLogPlot[data, opts]
```
Out[41]=



This plot shows a suggestive linear trend. Try fitting it to find the slope:

In[42]:=
```
logdata = Log[10, QuantityMagnitude[data]]
```
Out[42]=
```
{{10.7627, 6.8808447}, {11.0343, 7.28811836}, {11.1749, 7.49911152},
 {11.3578, 7.77345757}, {11.8912, 8.57328440}, {12.1543, 8.96815237},
 {12.458, 9.42347034}, {12.653, 9.71603830}, {12.7725, 9.895}}
```

In[43]:=
```
llfit[x_] = Fit[logdata, {1, x}, x]
```
Out[43]=
```
-9.26216 + 1.4999 x
```

The slope is very nearly 1.5.

In[44]:=
```
Show[{ListPlot[logdata], Plot[llfit[x], {x, logdata[[1, 1]], logdata[[-1, 1]]},
   PlotStyle → Red, PlotLegends → "Slope=1.49983"]}, LabelStyle → Larger,
 Frame → True, FrameLabel → {"Log(Mean Radius)", "Log(Orbital Period)"},
 ImageSize → Scaled[0.75], AspectRatio → 1/2
]
```
Out[44]=



Log-Log Plots and Power Laws

Suppose the period T is related to the orbital radius R by a power law $T = c\,R^m$. How would that show up in a plot? Try taking logs of both sides:

$T = c\,R^m$

$\log(T) = \log(c) + \log(R^m) = \log(c) + m\log(R)$

This suggests that if you plot log(T) vs. log(R), you will get a straight line with slope ‘m’. In this case, we have m = 1.5. This means

$T = c\,R^{1.5} = c\,R^{3/2}$

squaring both sides gives

$T^2 = c\,R^3$

which is Kepler’s 3rd law.

Kepler’s Third Law

The Log-Log analysis suggests that another way to plot it is to plot T^2 vs. R^3. It shows a linear trend, so we can fit it with a straight line. We will see that the slope is given by $\frac{4\pi^2}{G\,M_{sun}}$

In[45]:=
```
data32 = Table[{QuantityMagnitude[data[[i, 1]]]^3,
    QuantityMagnitude[data[[i, 2]]]^2}, {i, 1, Length[data]}];
```

In[46]:=
```
fit = LinearModelFit[data32, {x}, x, IncludeConstantBasis → False];
```

In[47]:=
```
Show[{ListPlot[data32],
  Plot[fit[x], {x, data32[[1, 1]], data32[[-1, 1]]}, PlotStyle → Red]},
 LabelStyle → Larger, Frame → True,
 FrameLabel → {"(Mean Radius (m))^3", "(Orbital Period (s))^2"},
 ImageSize → Scaled[0.75], AspectRatio → 1/2
]
```

Out[47]=



In[48]:=
```
slope = fit'[x]
```

Out[48]=

2.97414×10^{-19}

In[49]:=
```
G = Quantity[1, "GravitationalConstant"];
Msun = Quantity[1, "SolarMass"];
UnitConvert[4 π^2 / (G * Msun), "SIBase"]
```

Out[51]=

$2.975\times10^{-19}\ s^2/m^3$

It is useful to express the equation relating T^2 and R^3 in the following way:

$$GM_{\text{sun}}T^2 = 4\pi^2 R^3 \tag{1}$$

where $M_{\text{sun}} = 1.989 \times 10^{30}\,\text{kg}$ is the mass of the sun, and $G = 6.67 \times 10^{-11}\,\text{Nm}^2/\text{kg}^2$ is Newton's gravitational constant. This is also known as Kepler's Third Law. (More on that later).

From this, we can deduce the form of the gravitational force.

Newton's Law of Universal Gravitation

Consider again a planet moving at a speed v in a circular orbit of radius R. The period T is related to the speed v by $v = \dfrac{2\pi R}{T}$.



$$\begin{aligned} F &= ma \\ F &= m\frac{v^2}{R} = \frac{m}{R}\left(\frac{2\pi R}{T}\right)^2 \\ F &= m\frac{4\pi^2 R}{T^2} \end{aligned}$$

Now combine that with the experimentally-determined Kepler's Third Law:

$$\begin{aligned} GM_{\text{sun}}T^2 &= 4\pi^2 R^3 \\ T^2 &= 4\pi^2 R^3/GM_{\text{sun}} \end{aligned}$$

Plug that expression for T in to the force equation:

$$F = m\frac{4\pi^2 R}{4\pi^2 R^3/GM_{\text{sun}}} \tag{2}$$

$$F = \boxed{\frac{GM_{\text{sun}}m}{R^2}} \tag{3}$$

This is Newton's law of gravitation.

Vector Formulation

The gravitational force is a vector. It has magnitude and direction. The magnitude is given by Eq. 3. The direction is attractive. It is useful to develop a simple notation to indicate the direction of the force.

Consider two masses, M_1 and M_2. Assume that mass M_1 is at the origin, and mass M_2 is located at $\vec{r}$, a distance r away.



The force on M_2 is attractive and points back towards M_1. We can describe that in terms of a *unit vector* which we call $\hat{r}$. Specifically, define the unit vector (pronounced as "r hat") by

$$\hat{r} \equiv \frac{\vec{r}}{|\vec{r}|} = \frac{\vec{r}}{r}$$

This vector is dimensionless, and has a magnitude of 1. It's sole purpose is to point in the $\vec{r}$ direction. (We will never actually calculate an $\hat{r}$ in this course. It's sole purpose is to point in a direction, and we will simply draw the gravitational force vector on a free body diagram pointing in the correct direction and use that diagram to calculate any components neeeded.)

We can now write Newton's Law of Universal Gravitation in vector form by multiplying the magnitude by $(-\hat{r})$ to indicate that it points *towards* the other mass.



$$\boxed{\vec{F} = -\frac{GM_1M_2}{r^2}\hat{r}}$$

Key ideas

1. The force is always attractive between masses.
2. The force is proportional to $1/r^2$, and tends to zero as $r \to \infty$.
3. The gravitational force is a *vector*. It obeys the law of superposition. If there are two masses, m_1, and m_2 pulling on mass m_0, the total force on m_0 is the vector sum of the two forces:



4. One of Newton's key insights: This law is *universal*. Any two masses are attracted by this law. It applies to small objects on Earth as well as large objects (planets, stars, and galaxies) in space.

13.2 Weight

Force exerted on a 1 kg mass on the surface of Earth

Consider a 1.00 kg mass on the surface of the Earth. What is the magnitude of the gravitational force exerted by the Earth on that mass?



$$
\begin{aligned}
F &= \frac{GM_{\text{Earth}}m}{R^2_{\text{Earth}}} = m \times \frac{GM_{\text{Earth}}}{R^2_{\text{Earth}}} \\
G &= 6.67 \times 10^{-11}\,\text{Nm}^2/\text{kg}^2 \\
M_{\text{Earth}} &= 5.972 \times 10^{24}\,\text{kg} \\
m &= 1.00\,\text{kg} \\
R_{\text{Earth}} &= 6.367 \times 10^{6}\,\text{m} \;\; (\text{Note: not exactly spherical}) \\
F &= 9.83\,\text{N} = (1.00\,\text{kg}) \times (9.83\,\text{m/s}^2)
\end{aligned}
$$

This is where we get our value for g in $F = mg = m \times 9.8\,\text{m/s}^2$. (The Earth is rotating, and also not exactly spherical. These complicate the situation somewhat, but we will ignore those complications.)

Important Note: Outside a spherically symmetric object, gravity acts as if all the mass were concentrated at the center.

13.3 The Motion of Satellites

We can use Newton's laws to explore the orbits of satellites. Consider the International Space Station (ISS). It orbits the Earth in a roughly circular orbit at a height $h = 400\,\text{km} = 4.00 \times 10^5\,\text{m}$ above the surface of the earth. What is the period of its orbit?



Data:

$$\begin{aligned} h &= 4.00 \times 10^5\,\text{m} \\ R_E &= 6.371 \times 10^6\,\text{m} \\ r &= R_E + h = 6.771 \times 10^6\,\text{m} \\ M_E &= 5.972 \times 10^{24}\,\text{kg} \end{aligned}$$

The space station travels in a circular orbit a distance $r = R_E + h$ away from the center of the Earth. Apply Newton's Second law to find the speed, and then the orbital period.:

$$\begin{aligned} \sum F &= ma \\ \sum F &= m\frac{v^2}{r} \\ \frac{GM_e m}{r^2} &= m\frac{v^2}{r} \\ \frac{GM_e}{r} &= v^2 \\ v &= \sqrt{\frac{GM_e}{r}} \\ v &= \sqrt{\frac{(6.67 \times 10^{-11}\,\text{Nm}^2/\text{kg}^2) \times (5.972 \times 10^{24}\,\text{kg})}{6.771 \times 10^6\,\text{m}}} \\ v &= \boxed{7672\,\text{m/s}} \\ T &= \frac{2\pi r}{v} = \frac{2\pi \times (6.771 \times 10^6\,\text{m})}{7672\,\text{m/s}} \\ T &= \boxed{5545\,\text{s}} \approx \boxed{92\,\text{minutes}} \end{aligned}$$

What is the free-fall acceleration-, or the effective "g," at that height?

$$
\begin{aligned}
F &= ma \\
a &= \frac{F}{m} \\
a &= \frac{\frac{GM_E m}{r^2}}{m} \\
a &= \frac{GM_E}{r^2} \\
a &= \frac{(6.67 \times 10^{-11}\,\mathrm{Nm^2/kg^2}) \times (5.972 \times 10^{24}\,\mathrm{kg})}{(6.771 \times 10^{6}\,\mathrm{m})^2} \\
a &= 8.69\,\mathrm{m/s^2}
\end{aligned}
$$

This is only slightly less than on Earth's surface. Why do the astronauts feel weightless? They are continually falling!

Recap

We have two main ideas in this section: The first is Newton's second law:

$$\sum \vec{F} = m\vec{a}$$

The second is the Newton's law of universal gravitation: Any two masses M_1 and M_2, separated by a distance r are attracted by a force given by

$$\vec{F} = -\frac{GM_1M_2}{r^2}\hat{r}$$

This force law allows us to use $\sum \vec{F} = m\vec{a}$ in a richer variety of applications.

What's Next?

Energy

Orbits

Examples!